NAG Library Routine Document

G05RFF

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

1 Purpose

G05RFF generates pseudorandom uniform bivariates with joint distribution of a Frank Archimedean copula.

2 Specification

```
SUBROUTINE G05RFF (N, THETA, SORDER, STATE, X, LDX, SDX, IFAIL)
INTEGER            N, SORDER, STATE(*), LDX, SDX, IFAIL
REAL (KIND=nag_wp) THETA, X(LDX,SDX)
```

3 Description

Generates pseudorandom uniform bivariates $\{u_1, u_2\} \in [0,1]^2$ whose joint distribution is the Frank Archimedean copula C_θ with parameter θ, given by

$$C_\theta = -\frac{1}{\theta}\ln\left[1 + \frac{\left(e^{-\theta u_1} - 1\right)\left(e^{-\theta u_2} - 1\right)}{e^{-\theta} - 1}\right], \quad \theta \in (-\infty, \infty) \setminus \{0\}$$

with the special cases:

$C_{-\infty} = \max(u_1 + u_2 - 1, 0)$, the Fréchet–Hoeffding lower bound;

$C_0 = u_1 u_2$, the product copula;

$C_\infty = \min(u_1, u_2)$, the Fréchet–Hoeffding upper bound.

The generation method uses conditional sampling.

One of the initialization routines G05KFF (for a repeatable sequence if computed sequentially) or G05KGF (for a non-repeatable sequence) must be called prior to the first call to G05RFF.

4 References

Nelsen R B (2006) *An Introduction to Copulas* (2nd Edition) Springer Series in Statistics

5 Arguments

1: N – INTEGER *Input*

On entry: n, the number of bivariates to generate.

Constraint: N ≥ 0.

2: THETA – REAL (KIND=nag_wp) *Input*

On entry: θ, the copula parameter.

3: SORDER – INTEGER *Input*

On entry: determines the storage order of variates; the (i,j)th variate is stored in X(i,j) if SORDER $= 1$, and X(j,i) if SORDER $= 2$, for $i = 1, 2, \ldots, n$ and $j = 1, 2$.

Constraint: SORDER $= 1$ or 2.

4: STATE(∗) – INTEGER array *Communication Array*

Note: the actual argument supplied **must** be the array STATE supplied to the initialization routines G05KFF or G05KGF.

On entry: contains information on the selected base generator and its current state.

On exit: contains updated information on the state of the generator.

5: X(LDX, SDX) – REAL (KIND=nag_wp) array *Output*

On exit: the n bivariate uniforms with joint distribution described by C_θ, with X(i, j) holding the ith value for the jth dimension if SORDER $= 1$ and the jth value for the ith dimension if SORDER $= 2$.

6: LDX – INTEGER *Input*

On entry: the first dimension of the array X as declared in the (sub)program from which G05RFF is called.

Constraints:

if SORDER $= 1$, LDX $\geq$ N;
if SORDER $= 2$, LDX ≥ 2.

7: SDX – INTEGER *Input*

On entry: the second dimension of the array X as declared in the (sub)program from which G05RFF is called.

Constraints:

if SORDER $= 1$, SDX ≥ 2;
if SORDER $= 2$, SDX $\geq$ N.

8: IFAIL – INTEGER *Input/Output*

On entry: IFAIL must be set to 0, -1 or 1. If you are unfamiliar with this argument you should refer to Section 3.4 in How to Use the NAG Library and its Documentation for details.

For environments where it might be inappropriate to halt program execution when an error is detected, the value -1 or 1 is recommended. If the output of error messages is undesirable, then the value 1 is recommended. Otherwise, if you are not familiar with this argument, the recommended value is 0. **When the value -1 or 1 is used it is essential to test the value of IFAIL on exit.**

On exit: IFAIL $= 0$ unless the routine detects an error or a warning has been flagged (see Section 6).

6 Error Indicators and Warnings

If on entry IFAIL $= 0$ or -1, explanatory error messages are output on the current error message unit (as defined by X04AAF).

Errors or warnings detected by the routine:

IFAIL $= 1$

On entry, corrupt STATE argument.

IFAIL $= 3$

On entry, N $= \langle value \rangle$.
Constraint: N ≥ 0.

IFAIL $= 4$

On entry, invalid SORDER.
Constraint: SORDER $= 1$ or 2.

IFAIL $= 6$

On entry, LDX must be at least $\langle value \rangle$: LDX $= \langle value \rangle$.

IFAIL $= 7$

On entry, SDX must be at least $\langle value \rangle$: SDX $= \langle value \rangle$.

IFAIL $= -99$

An unexpected error has been triggered by this routine. Please contact NAG.

See Section 3.9 in How to Use the NAG Library and its Documentation for further information.

IFAIL $= -399$

Your licence key may have expired or may not have been installed correctly.

See Section 3.8 in How to Use the NAG Library and its Documentation for further information.

IFAIL $= -999$

Dynamic memory allocation failed.

See Section 3.7 in How to Use the NAG Library and its Documentation for further information.

7 Accuracy

Not applicable.

8 Parallelism and Performance

G05RFF is threaded by NAG for parallel execution in multithreaded implementations of the NAG Library.

Please consult the X06 Chapter Introduction for information on how to control and interrogate the OpenMP environment used within this routine. Please also consult the Users' Note for your implementation for any additional implementation-specific information.

9 Further Comments

In practice, the need for numerical stability restricts the range of θ such that:

if $\theta < \ln \epsilon_s$, the routine returns pseudorandom uniform variates with $C_{-\infty}$ joint distribution;

if $|\theta| < 1.0 \times 10^{-6}$, the routine returns pseudorandom uniform variates with C_0 joint distribution;

if $\theta > \ln \epsilon$, the routine returns pseudorandom uniform variates with C_{∞} joint distribution;

where ϵ_s is the safe-range parameter, the value of which is returned by X02AMF; and ϵ is the ***machine precision*** returned by X02AJF.

10 Example

This example generates thirteen variates for copula $C_{-12.0}$.

10.1 Program Text

```
    Program g05rffe

!     G05RFF Example Program Text

!     Mark 26 Release. NAG Copyright 2016.

!     .. Use Statements ..
      Use nag_library, Only: g05kff, g05rff, nag_wp, x04caf
!     .. Implicit None Statement ..
      Implicit None
!     .. Parameters ..
      Integer, Parameter               :: lseed = 1, nin = 5, nout = 6
!     .. Local Scalars ..
      Real (Kind=nag_wp)               :: theta
      Integer                          :: genid, ifail, ldx, lstate, n, sdx,   &
                                          sorder, subid
!     .. Local Arrays ..
      Real (Kind=nag_wp), Allocatable  :: x(:,:)
      Integer                          :: seed(lseed)
      Integer, Allocatable             :: state(:)
!     .. Executable Statements ..
      Write (nout,*) 'G05RFF Example Program Results'
      Write (nout,*)
      Flush (nout)

!     Skip heading in data file
      Read (nin,*)

!     Read in the base generator information and seed
      Read (nin,*) genid, subid, seed(1)

!     Initial call to initializer to get size of STATE array
      lstate = 0
      Allocate (state(lstate))
      ifail = 0
      Call g05kff(genid,subid,seed,lseed,state,lstate,ifail)

!     Reallocate STATE
      Deallocate (state)
      Allocate (state(lstate))

!     Initialize the generator to a repeatable sequence
      ifail = 0
      Call g05kff(genid,subid,seed,lseed,state,lstate,ifail)

!     Read in sample size and order
      Read (nin,*) n, sorder

      If (sorder==1) Then
!       X(N,2)
        ldx = n
        sdx = 2
      Else
!       X(2,N)
        ldx = 2
        sdx = n
      End If
      Allocate (x(ldx,sdx))

!     Read in parameter
      Read (nin,*) theta

!     Generate variates
      ifail = 0
      Call g05rff(n,theta,sorder,state,x,ldx,sdx,ifail)

!     Display the variates
      If (sorder==1) Then
!       X(N,2)
```

```
        ifail = 0
        Call x04caf('General',' ',n,2,x,ldx,                                &
          'Uniform variates with copula joint distribution',ifail)
      Else
!       X(2,N)
        ifail = 0
        Call x04caf('General',' ',2,n,x,ldx,                                &
          'Uniform variates with copula joint distribution',ifail)
      End If

    End Program g05rffe
```

10.2 Program Data

```
G05RFF Example Program Data
1  1  1762543          :: GENID,SUBID,SEED(1)
13  1                  :: N,SORDER
-12.0                  :: THETA
```

10.3 Program Results

```
 G05RFF Example Program Results

 Uniform variates with copula joint distribution
           1       2
  1   0.6364  0.1411
  2   0.1065  0.8967
  3   0.7460  0.1843
  4   0.7983  0.1254
  5   0.1046  0.9982
  6   0.4925  0.6901
  7   0.3843  0.6250
  8   0.7871  0.1654
  9   0.4982  0.5298
 10   0.6717  0.2902
 11   0.0505  0.9554
 12   0.2580  0.8190
 13   0.6238  0.3014
```